
Introduzione 
 

L’orologio digitale GPS, nasce dall’esigenza di sincronizzare i dati sismici registrati nella sala 
operativa dell’Osservatorio “V.Nigri” di Foggia. Nell’area geografica del foggiano ci sono attualmente 
7 stazioni sismiche analogiche, che trasmettono i dati tramite linee telefoniche CDA verso 
l’Osservatorio. L’acquisizione delle forme d’onda avviene esclusivamente su carta termosensibile 
tramite helicorder. Ripercorriamo brevemente e senza pretesa di completezza, le varie fasi che hanno 
contraddistinto l’evoluzione della tacca marcatempo nell’osservatorio. In passato tale struttura si è 
avvalsa di strumentazione autocostruita dal padre fondatore, Prof. Vincenzo Nigri.  

 

 
 

Figura 1. Parte del sismogramma del 16-01-1981 Basilicata e Lucania avvertito a Foggia. È possibile 
vedere la marca del tempo; l’intervallo tra le due frecce corrisponde a un minuto. 

 
Dai sismogrammi storici è possibile vedere la tacca del marcatempo (Figura 1) che veniva 

prodotta attraverso metodi esclusivamente meccanici. Il sistema di scrittura a pennino su carta 
affumicata veniva collegato ad un orologio a carica manuale, opportunamente connesso con apposite 
leve agli ingranaggi dell’ora. Il risultato portava ad evidenti svantaggi sia in termini di precisione che di 
funzionalità. 

Successivamente, con l’avanzata tecnologica ed il conseguente cambiamento radicale della 
strumentazione impiegata sia per la registrazione del segnale sismico che per la sua rilevazione, è stato 
necessario rivedere il sistema della marcatura temporale abbandonando tecniche meccaniche per passare 
a quelle elettroniche. I primi esperimenti adottavano come riferimento quello del DCF-77 il noto 
segnale trasmesso con una potenza di 50kW da Mainflingen, vicino a Francoforte, ma la cattiva 
ricezione del medesimo, a causa sia di problemi all’impianto radio che della scarsa potenza del segnale 
a queste latitudini, fu presto abbandonato. Al suo posto fu impiegato per diversi anni il segnale orario 
codificato (SRC), che viene generato dall’IEN (Istituto Elettrotecnico Nazionale) e costituisce uno dei 



 
 

6 

sistemi di riferimento utilizzati per disseminare in tempo reale in Italia la scala di tempo nazionale UTC 
(IEN) e che viene diffuso mediante le reti radiofoniche della RAI (Radio Televisione Italiana). 

Il segnale SRC veniva ricevuto con una normale radio AM ed, opportunamente demodulato, 
andava a sincronizzare un orologio programmatore (identico a quello ancora in funzione nella sala 
operativa della sede di Roma dell’ INGV) da cui veniva generato con cadenza minuti-ore- mezzanotte, 
un opportuno segnale che pilotava un relè che chiudeva la tensione per la generazione della marca del 
tempo sui rulli. Questo sistema aveva bisogno di regolazioni periodiche per risincronizzare l’orologio 
programmatore con il tempo UTC trasmesso. Allo scopo di evitare delle inevitabili imprecisioni dovute 
alla fase di sincronizzazione dell’orologio da parte dell’utente, abbiamo ritenuto opportuno progettare 
un orologio GPS che non richiede regolazioni in quanto automatico. La descrizione della progettazione 
e realizzazione di questo dispositivo è della presente pubblicazione. 
 

 
 

Figura 2. Sismogramma ottenuto con l’orologio sincronizzato GPS. 

1.  Descrizione del progetto 
 

L’orologio è costituito da tre schede: una si occupa dell’acquisizione dei dati GPS, generazione 
del marcatempo e aggiornamento del display, una è il ricevitore GPS (della Motorola), e l’ultima 
fornisce la giusta alimentazione a tutto il sistema (vedi schemi a pagg.12, 13, 14). La scheda di 
acquisizione si basa sull’impiego del microprocessore Dallas 80C320 unitamente ad una RAM esterna, 
una GAL per la codifica degli indirizzi ed a una flash-eprom per gli aggiornamenti del firmware e la 
diagnostica dei registri del processore, tramite un programma residente (DIAG320).  

L’80C320 è dotato di due porte seriali utilizzate per la ricezione dei dati provenienti dal ricevitore 
GPS con un bit-rate di 9600 bit/sec (seriale1) e per la diagnostica del sistema od eventuali 
aggiornamenti al firmware a 19200 bit/s (seriale0). Nella figura 5 è riportato il diagramma di flusso del 
programma assembler (pag. 17) per l’80C320 e qui di seguito se ne spiegano i passi principali. La prima 
operazione consiste nella programmazione del ricevitore GPS tramite l’invio di un comando nel formato 
proprietario della Motorola (Motorola-Binary-Format - vedi pagina 15 per la sua descrizione), 
consistente in una serie di 8 byte inviati sulla seriale1: 
 

@@Eamc<CR><LF> 
 

I primi quattro byte sono l’identificativo del comando, m determina il modo di risposta, ossia ogni 
quanti secondi si vogliono ricevere i dati dal GPS, ed il suo range è 0...255; se m=0 allora i dati vengono 
inviati una volta sola, in risposta al comando inviato (polling mode) mentre per tutti gli altri valori 
(continuous mode) vengono inviati ogni m secondi. Nel nostro caso m=1; c rappresenta il checksum dei 
byte precedenti (cioè di @@Eam) seguito da un carriage-return (<CR>) ed un line-feed (<LF>). 
Appena il GPS ha ricevuto questo comando comincia ad inviare 74 byte di informazioni, relative alla 
data, ora, posizione, stato dei satelliti ecc... Le informazioni inviate sono sempre precedute dai byte 
@@Ea, ossia dalla ripetizione del codice comando che le ha richieste. Sfruttando questa caratteristica si 
esegue un controllo sui byte ricevuti e solo quando i caratteri @@ vengono riconosciuti, comincia la 
memorizzazione in RAM, a partire dalla locazione 05020h. Prima che i dati GPS vengano scritti nella 
RAM, il processore inizializza il display, formato da due righe di 40 caratteri ciascuna e alloca in 
memoria tutti i caratteri da scrivere che sono fissi, come gli spazi tra due parole o i due punti tra le ore 
ed i minuti (fig.3). 
 
 

 
Figura 3. Display LCD dell’orologio GPS. Sulla seconda riga a destra si trovano le informazioni sul 
numero dei satelliti visibili (satv) e quelli agganciati (satt).  
 

A questo punto viene inviata al display la scritta OSSERVATORIO “NIGRI” FOGGIA che 
occupa tutta la prima riga. La seconda riga invece, contiene dei campi (come l’orario) variabili, i quali 
sono aggiornati in seguito all’acquisizione dei dati GPS, infatti appena memorizzati in RAM vengono 
convertiti dal formato binario a quello ascii (routine bin-to-ascii) ed inviati sull’LCD tramite il bus dati 
dell’80C320. Nel frattempo viene controllato se sulla seriale0 è stato trasmesso qualche carattere, da 
parte di un eventuale utente che volesse interagire con l’orologio, ad esempio per eseguire una 
diagnostica di funzionamento; infatti è possibile fare ciò, collegandosi al processore, su tale seriale, con 
un PC ed un comune programma per la gestione delle porte seriali, ad esempio Procomm o il Terminal 
di Windows. Basterà inviare il carattere “F” ed il normale flusso di programma si bloccherà, lasciando il 
controllo al programma residente nella flash-eprom: il DIAG320. Appena collegati si vedranno 
comparire sullo schermo i secondi in avanzamento provenienti dall’orario GPS allo scopo di rilevare 



 
 

8 

delle interruzioni sul normale andamento del programma in esecuzione nel processore. Ciò risulta molto 
importante per eseguire delle operazioni di manutenzione sull’apparato circuitale, nel momento in cui si 
deve operare sul circuito senza disporre del display LCD, situazione che si potrebbe verificare nel 
momento in cui venga spedita alla sede centrale di Roma la sola scheda processore per l’eventuale 
manutenzione o per effettuare un aggiornamento firmware. L’ambiente DIAG320 è un utile strumento 
per effettuare delle diagnosi a basso livello sul processore, permettendo ad esempio la visione del 
contenuto di alcuni registri oppure per effettuare il download di eventuali aggiornamenti firmware. Tale 
strumento è stato pensato esclusivamente per essere usato da personale qualificato e non dall’utente 
dell’apparato, dal momento che è facile creare malfunzionamenti se non correttamente usato, pertanto 
una sua approfondita descrizione esula dagli scopi del presente lavoro. Da tale ambiente si esce con il 
comando G1000. In assenza di caratteri ricevuti o differenti da “F”, il processore continuerà ad eseguire 
il programma principale, ritornando ad acquisire i dati relativi al secondo successivo. La routine del 
marcatempo è eseguita appena i dati relativi all’orario sono disponibili, prima di qualsiasi scrittura sul 
display, al fine di evitare pericolosi ritardi temporali nella generazione dalla marca del tempo. 

Come già detto, l’hardware dell’orologio GPS è suddiviso su tre schede (figg.6 e 7). La scheda di 
acquisizione dati e quella dell’alimentazione sono alloggiate all’interno di un unico chassis contenente 
anche il display LCD e le uscite del relé per il marcatempo; la scheda che ospita il ricevitore GPS con 
l’antenna e l’interfaccia per l’adattamento dei segnali (da RS-232 a TTL e viceversa), è montata sul 
terrazzo dell’osservatorio (fig.8), a circa 40m di distanza dall’acquisitore dal momento che quest’ultimo 
si trova in un seminterrato da dove è impossibile portare all’esterno l’antenna GPS (lunga 6m) con 
buoni risultati per la ricezione dei satelliti a causa della zona d’ombra creata dall’edificio sovrastante. Il 
GPS è alimentato dal medesimo cavo che porta i dati (formato RS-232) all’acquisitore; per ovviare al 
problema di eventuali cadute di tensione, vista la lunghezza del cavo (40m), al ricevitore viene inviata la 
tensione del condensatore di livellamento dell’alimentatore (circa 17V) e successivamente abbassata a 
5V da un DC-DC converter (LM2825) montato sulla scheda che ospita il GPS. 

 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 

 
Figura 4. Orologio montato all’interno dell’osservatorio. 

Figura 5. Diagramma di flusso del programma per l’orologio GPS. 

Figura 6. Alimentatore (sinistra) e 
scheda processore (destra). In alto 
sulla seconda scheda sono visibili le 
due seriali usate per l’acquisizione e 
l’eventuale diagnostica da parte 
dell’utente. 

Figura 7. Scheda GPS e suo 
schema a blocchi. Il ricevitore 
è montato sulla scheda tramite 
una connessione 10 poli dual-
in-line. Il tutto è alloggiato 
all’interno di un contenitore 
stagno (IP65). 

 

Figura 8. Montaggio della scheda GPS 
sul terrazzo dell’osservatorio. Il cavo 
corrugato bianco racchiude l’antenna 
GPS, posizionata sopra una lastra 
metallica, che agisce da schermo 
riflettente e da riparo per la scheda 
sottostante.   

2.  Schema elettrico acquisitore 

3.  Schema elettrico alimentatore 

4.  Schema elettrico GPS 

5.  Descrizione del comando per la programmazione del GPS 

6.  Programma assembler CLOCK ASM 
 

$debug 
$xref 
$nomod51                  ;disattiva i parametri 8051 
$include(reg320.pdf)      ;attiva i parametri 80320 
 
;---------------------- Inizializzazioni ------------------------ 
;nome programma:clock.asm  
;data: 18/01/02 
 
        gps_ini    equ 05020h    ;inizio zona memoria per dati gps 
        gps_m      equ 05024h    ;locazione XRAM mese 
        gps_d      equ 05025h    ;locazione XRAM data 
        gps_y_h    equ 05026h    ;locazione XRAM anno prime due cifre 
        gps_y_l    equ 05027h    ;locazione XRAM anno seconde due cifre 
        gps_h      equ 05028h    ;locazione XRAM ora 
        gps_min    equ 05029h    ;locazione XRAM minuti 
        gps_s      equ 0502Ah    ;locazione XRAM secondi 
        gps_vis    equ 05046h    ;locazione satelliti visibili 
        gps_trk    equ 05047h    ;locazione XRAM satelliti trakkati 
        reg_rit    equ 20h      ; 
        reg_rit1   equ 21h      ; 
        reg_rit2   equ 22h      ; 
        xxx1       equ 23h      ; 
        xxx2       equ 24h      ; 
        bta1       equ 25h      ; 
        bta2       equ 26h      ; 
        bta3       equ 27h      ; 
        flag_m     equ 28h 
        flag_h     equ 29h 
        flag_g     equ 2ah 
        anno_v     equ 05200h   ;anno vero 
        anno_dif   equ 05201h   ;anno differenza 
;************inizio maschera********** 
        day_h      equ 05150h    ;giorno1 
        day_l      equ 05151h    ;giorno2 
        trat1      equ 05152h    ;separatore tratto 
        m_h        equ 05153h    ;prima cifra mese 
        m_l        equ 05154h    ;seconda cifra mese 
        trat2      equ 05155h    ;separatore tratto 
        yy1        equ 05156h    ;prima cifra anno 
        yy2        equ 05157h    ;seconda cifra anno 
        yy3        equ 05158h    ;terza cifra anno 
        yy4        equ 05159h    ;quarta cifra anno 
        null1      equ 0515ah 
        null2      equ 0515bh    ;separatore tratto 
        null3      equ 0515ch 
        null4      equ 0515dh 
        null5      equ 0515eh 
        hour_h     equ 0515fh    ;prima cifra ora 
        hour_l     equ 05160h    ;seconda cifra ora 
        punti1     equ 05161h    ;separatore ora 
        min_h      equ 05162h    ;prima cifra minuti 
        min_l      equ 05163h    ;seconda cifra minuti 
        punti2     equ 05164h    ;separatore ora 

        sec_h      equ 05165h    ;prima cifra secondi 
        sec_l      equ 05166h    ;seconda cifra secondi 
        null6      equ 05167h    ;spazio 
        null7      equ 05168h    ; 
        null8      equ 05169h    ;                scritta 
        s_1        equ 0516ah    ;                'numero' 
        a_1        equ 0516bh    ; 
        t_1        equ 0516ch    ; 
        v_1        equ 0516dh    ; 
        punti3     equ 0516eh    ;spazio 
        satv_h     equ 0516fh    ; 
        satv_l     equ 05170h    ; 
        null9      equ 05171h    ;                scritta 
        s_2        equ 05172h    ;              'satelliti' 
        a_2        equ 05173h    ; 
        t_2        equ 05174h    ; 
        t_3        equ 05175h    ; 
        punti4     equ 05176h 
        trk        equ 05177h    ; 
 
 
 
        org 1000h 
        ljmp begin 
 
versione:db 'clock16     ' 
 
 
gps_p:  db 40h,40h,45h,61h,01h,25h,0dh,0ah      ;inizializzazione 
        db 40h,40h,41h,77h,01h,37h,0dh,0ah      ;      GPS 
 
;---------------------- Programmazione GPS ------------------------ 
 
        org 1100h 
begin:  mov r0,#16d     ;Lunghezza dati GPS dell'inizializzazione 
        mov dptr,#gps_p ; 
 
gps1:   mov a,#00h 
        movc a,@a+dptr    ;programmazione gps 
        inc dptr 
        jnb 0c1h,$ 
        clr 0c1h 
        mov 0c1h,a 
        djnz r0,gps1 
 
        mov sp,#50h 
 
;---------------------------- Tabella DISPLAY -------------------------- 
 
        mov dptr,#null1 
        mov a,#' ' 
        movx @dptr,a 
 
        mov dptr,#trat1 
        mov a,#'-' 
        movx @dptr,a 

        mov dptr,#trat2 
        mov a,#'-' 
        movx @dptr,a 
 
        mov dptr,#null2 
        mov a,#' ' 
        movx @dptr,a 
 
        mov dptr,#null3 
        mov a,#' ' 
        movx @dptr,a 
 
        mov dptr,#null4 
        mov a,#' ' 
        movx @dptr,a 
 
        mov dptr,#null5 
        mov a,#' ' 
        movx @dptr,a 
 
        mov dptr,#punti1 
        mov a,#':' 
        movx @dptr,a 
 
 
        mov dptr,#punti2 
        mov a,#':' 
        movx @dptr,a 
 
        mov dptr,#punti3 
        mov a,#':' 
        movx @dptr,a 
 
        mov dptr,#punti4 
        mov a,#':' 
        movx @dptr,a 
 
        mov dptr,#null6 
        mov a,#' ' 
        movx @dptr,a 
 
        mov dptr,#null7 
        mov a,#' ' 
        movx @dptr,a 
 
        mov dptr,#null8 
        mov a,#' ' 
        movx @dptr,a 
 
        mov dptr,#null9 
        mov a,#' ' 
        movx @dptr,a 
 
        mov dptr,#s_1 
        mov a,#'s' 



 
 

19 

        movx @dptr,a 
 
        mov dptr,#a_1 
        mov a,#'a' 
        movx @dptr,a 
 
        mov dptr,#t_1 
        mov a,#'t' 
        movx @dptr,a 
 
        mov dptr,#v_1 
        mov a,#'v' 
        movx @dptr,a 
 
        mov dptr,#s_2 
        mov a,#'s' 
        movx @dptr,a 
 
        mov dptr,#a_2 
        mov a,#'a' 
        movx @dptr,a 
 
        mov dptr,#t_2 
        mov a,#'t' 
        movx @dptr,a 
 
        mov dptr,#t_3 
        mov a,#'t' 
        movx @dptr,a 
 
 
 
;--------- Routine di inizializzazione marcatempo --------------- 
 
        clr p3.4 
        mov flag_m,#00h 
        mov flag_h,#00h 
        mov flag_g,#00h 
 
;--------- Routine di inizializzazione display caratteri --------------- 
 
        mov dptr,#0c030h  ;indirizzo instruction register 
        mov a,#38h        ;set 8 bit,caratteri 5x7 
        lcall writ 
        mov a,#0ch        ;display on e cursor off 
        lcall writ 
        mov a,#06h        ;cursor shift to the right 
        lcall writ 
        mov a,#01         ;display clear 
        lcall writ 
        lcall rit 
 
        lcall present 
 
        jmp val_ini_3 
 



 
 

20 

msg: 
db '      OSSERVATORIO "NIGRI" FOGGIA       ' 
 
ritardo: 
        mov reg_rit2,#01h 
ritardo1: 
        mov reg_rit1,#01fh 
ritardo2: 
        mov reg_rit,#0ffh 
        djnz reg_rit,$ 
        djnz reg_rit1,ritardo2 
        djnz reg_rit2,ritardo1 
        ret 
 
rit: 
        mov reg_rit,#80h 
rit1: 
        mov reg_rit1,#01fh 
        djnz reg_rit1,$ 
        djnz reg_rit,rit1 
        ret 
 
writ:   movx @dptr,a         
        mov reg_rit,#0ffh 
        djnz reg_rit,$ 
        ret 
 
present: 
        mov reg_rit2,#40d 
        mov dptr,#msg 
        mov xxx1,dph 
        mov xxx2,dpl 
prese:  mov dph,xxx1 
        mov dpl,xxx2 
        mov a,#00h 
        movc a,@a+dptr     
        inc dptr 
        mov xxx1,dph 
        mov xxx2,dpl 
        mov dptr,#0c031h 
        lcall writ 
        lcall rit 
        djnz reg_rit2,prese 
        ret 
 
riga_2: 
        mov reg_rit2,#40d 
        mov dptr,#day_h 
        mov xxx1,dph 
        mov xxx2,dpl 
prese1: mov dph,xxx1 
        mov dpl,xxx2 
        movx a,@dptr 
        inc dptr 
        mov xxx1,dph 
        mov xxx2,dpl 

        mov dptr,#0c031h 
        lcall writ 
        lcall rit 
        djnz reg_rit2,prese1 
        ret 
 
;------------ Acquisizione dati GPS ---------------------- 
val_ini_3: 
        mov 60h,#0ffh 
kkk:    jnb 0c0h,$ 
        clr 0c0h 
        djnz 60h,kkk 
 
val_ini: 
 
        mov dptr,#gps_ini 
prel_dati: 
        jnb 0c0h,$ 
        clr 0c0h 
        mov a,0c1h 
        cjne a,#'@',prel_dati 
        movx @dptr,a 
        inc dptr 
        jnb 0c0h,$ 
        clr 0c0h 
        mov a,0c1h 
        cjne a,#'@',val_ini 
        movx @dptr,a 
        inc dptr 
 
        mov r0,#74d 
 
prel_dati1: 
 
        jnb 0c0h,$ 
        clr 0c0h 
        mov a,0c1h 
        movx @dptr,a 
        inc dptr 
        djnz r0,prel_dati1 
        jnb ri,ricom 
        clr ri 
        mov a,sbuf 
        cjne a,#'F',ricom       ;serve per uscire dal programma attendendo 
                                ;il tasto SHIFT F 
        ljmp 2000h 
 
ricom:  mov dptr,#gps_d         ; 
        movx a,@dptr            ;richiamo routine per conversione data 
        lcall bin_to_ascii      ; 
 
        mov dptr,#day_h         ; 
        mov a,bta2              ; 
        movx @dptr,a            ; 
        mov dptr,#day_l         ; 
        mov a,bta3              ; 

        movx @dptr,a            ; 
 
        mov dptr,#gps_m 
        movx a,@dptr 
        lcall bin_to_ascii 
        mov dptr,#m_h 
        mov a,bta2 
        movx @dptr,a 
        mov dptr,#m_l 
        mov a,bta3 
        movx @dptr,a 
 
        mov dptr,#yy1 
        mov a,#'2' 
        movx @dptr,a 
        mov dptr,#yy2 
        mov a,#'0' 
        movx @dptr,a 
 
 
 
        mov dptr,#gps_y_l 
        movx a,@dptr 
        mov r0,#0D2h 
        subb a,r0 
        add a,#02h 
        lcall bin_to_ascii 
        mov dptr,#yy3 
        mov a,bta2 
        movx @dptr,a 
        mov dptr,#yy4 
        mov a,bta3 
        movx @dptr,a 
 
        mov dptr,#gps_h 
        movx a,@dptr 
        lcall bin_to_ascii 
        mov dptr,#hour_h 
        mov a,bta2 
        movx @dptr,a 
        mov dptr,#hour_l 
        mov a,bta3 
        movx @dptr,a 
 
        mov dptr,#gps_min 
        movx a,@dptr 
        lcall bin_to_ascii 
        mov dptr,#min_h 
        mov a,bta2 
        movx @dptr,a 
        mov dptr,#min_l 
        mov a,bta3 
        movx @dptr,a 
 
        mov dptr,#gps_s 
        movx a,@dptr 

        lcall bin_to_ascii 
        mov dptr,#sec_h 
        mov a,bta2 
        movx @dptr,a 
        mov dptr,#sec_l 
        mov a,bta3 
        movx @dptr,a 
        jnb ti,$ 
        clr ti 
        mov sbuf,a 
 
        mov dptr,#gps_vis 
        movx a,@dptr 
        lcall bin_to_ascii 
        mov dptr,#satv_h 
        mov a,bta2 
        movx @dptr,a 
        mov dptr,#satv_l 
        mov a,bta3 
        movx @dptr,a 
 
 
        mov dptr,#gps_trk 
        movx a,@dptr 
        lcall bin_to_ascii 
        mov dptr,#trk 
        mov a,bta3 
        movx @dptr,a 
        lcall marcatempo 
        lcall riga_2 
        lcall present 
        ljmp val_ini 
 
;------------------- Conversione binaria ad ASCII ------------------- 
 
bin_to_ascii: 
        mov b,#100d        ;divisione del numero per 100 decimale 
        div ab             ;l'intero e' in (A) ed il resto in (B) 
        add a,#30h         ;trasformazione in numero ASCII 
        mov bta1,a         ;trasferimento nel registro di uscita 
        mov a,b            ;resto da dividere in (A) 
        mov b,#10d         ;divisione del resto per 10 decimale 
        div ab             ;l'intero e' in (A) ed il resto in (B) 
        add a,#30h         ;trasformazione in numero ASCII 
        mov bta2,a         ;trasferimento nel registro di uscita 
        mov a,b            ;resto in (A) 
        add a,#30h         ;trasformazione in numero ASCII 
        mov bta3,a         ;trasferimento nel registro di uscita 
        ret                ;fine routine.  
 
marcatempo: 
              mov dptr,#gps_s 
              movx a,@dptr 
              cjne a,#00h,marca_esci 
              mov dptr,#gps_min 
              movx a,@dptr 

              cjne a,#00h,marca_1sec 
              mov dptr,#gps_h 
              movx a,@dptr 
              cjne a,#00h,marca_4sec 
 
              setb p3.4               ;scrive marca tempo da 8 secondi 
              mov flag_g,#0ffh 
 
 
              jmp marca_esci 
 
 
marca_4sec:      setb p3.4       ;scrive marca tempo da 4 secondi 
                 mov flag_h,#0ffh 
                 jmp marca_esci 
 
marca_1sec:      setb p3.4       ;scrive marca tempo da 1 secondi 
                 mov flag_m,#0ffh 
                 jmp marca_esci 
 
 
marca_esci:     mov a,flag_m 
                cjne a,#0ffh,esci_m 
                mov dptr,#gps_s 
                movx a,@dptr 
                 
 
                cjne a,#01h,esci_g 
                clr p3.4 
                mov flag_m,#00h 
 
esci_m:         mov a,flag_h 
                cjne a,#0ffh,esci_h 
                mov dptr,#gps_s 
                movx a,@dptr 
                 
                cjne a,#04h,esci_g 
                clr p3.4 
                mov flag_h,#00h 
 
esci_h:         mov a,flag_g 
                cjne a,#0ffh,esci_g 
              mov dptr,#gps_s 
              movx a,@dptr 
                 
                cjne a,#08h,esci_g 
                clr p3.4 
                mov flag_g,#00h 
 
esci_g:         ret 
 
                end 

7.  Programma in ABEL implementato nell GAL22V10 
 

MODULE clock// programma per gal22v10 plcc  
//file clock.abl ver. del 17/01/02 ore 9:00 
// programma di controllo per orologio GPS con dispaly 
 
a15 pin 5; 
a14 pin 6; 
a13 pin 7;  
H,L,X = 1,0,.X.; 
Address = [a15,a14,a13,X,X,X,X,X,X,X,X,X,X,X,X,X]; 
 
!pgm pin 2; 
!rd pin 9;  
!wr pin 10; 
!psen pin 11; 
cs_f pin 17; 
cs_d pin 18; 
rd_r pin 19; 
wr_r pin 20; 
wr_f pin 21; 
rd_f pin 23; 
LBE pin 27; 
HBE pin 25; 
 
 
equations 
!cs_f = (Address <= ^h3fff) # ((Address > ^H3fff) & pgm); 
!rd_f = !pgm & psen; 
!wr_f = pgm & wr; 
!rd_r = (psen & (Address >= ^H4000)) # rd; 
!wr_r = !pgm & wr; 
!cs_d = a15 & !a14 & a13 & wr ; //a000h (display) 
 
END 

8. Manuale orologio GPS Quick Start per l’utente 

